

# Complex numbers in Maple (I, evalc, etc..)

You will undoubtedly have encountered some complex numbers in Maple long before you begin studying them seriously in Math 241. For example, solving polynomial equations often leads to complex numbers:

```
> solve(x^2+3*x+11=0,x);
```

$$-\frac{3}{2} + \frac{1}{2}I\sqrt{35}, -\frac{3}{2} - \frac{1}{2}I\sqrt{35}$$

Maple uses a capital I to represent the square root of -1 (commonly called i in mathematics texts and sometimes called j in engineering texts). You use I to input complex numbers as well:

```
> z:=4+5*I; w:=3-2*I;
```

$$z := 4 + 5 I$$

$$w := 3 - 2 I$$

All of the basic arithmetic and standard functions work on complex numbers -- so we can add, subtract, multiply, divide, take exponentials, sines, Bessel functions, etc.. of complex numbers:

```
> z+w; z^2; z/w; exp(z);
```

$$7 + 3 I$$

$$-9 + 40 I$$

$$\frac{2}{13} + \frac{23}{13} I$$

$$e^{(4+5 I)}$$

To force Maple to report a complex number in "a+bI" format, there is the command evalc (for "evaluate as a complex number" -- it is similar to evalf which forces decimal, or floating point, evaluation):

```
> evalc(exp(z));
```

$$e^4 \cos(5) + I e^4 \sin(5)$$

A few other complex functions Maple knows about are: `conjugate` , `argument` , and `abs` . They do the obvious things:

```
> conjugate(z); argument(w); abs(z);
```

$$4 - 5 I$$
$$-\arctan\left(\frac{2}{3}\right)$$
$$\sqrt{41}$$

The absolute value in this case is the complex absolute value (or modulus -- the distance from the complex number to the origin in the Argand plane).

Maple understands complex limits -- to calculate them, use the "complex" option in the `limit` command (we cleared the value of `z` before executing the following):

```
> limit((exp(I*z)-1)/z, z=0, complex);
```

$I$

Complex derivatives work the same in Maple as ordinary real derivatives -- integrals are contour integrals, of course , which require parametrization of the contour and substitution of the parametrization into `dz`. For examples of this see the demonstration `m241demo9` and the section of this manual containing answers to selected Math 241 problems.


Finally, for exploring the geometry of complex functions as mappings from the plane to itself (or of the `z`-plane to the `w`-plane), there is the `conformal` command in the "plots" library; load it by typing:

```
> with(plots, conformal);
```

`[conformal]`

The `conformal` command illustrates what happens to lines parallel to the coordinate axes when a complex function is applied. For example:


```
> conformal(exp(z), z=-3-Pi*I..1+Pi*I);
```



Note that this is the transformation from rectangular to polar coordinates. The syntax of the `conformal` command is: first, you supply the complex expression that is to be applied, and then give the domain -- the domain is always a rectangle in the complex plane, and you need to supply the (complex) names of the points at the lower left corner ( $-3 - \pi i$  in this case) and the upper right corner ( $1 + \pi i$  in this case) of the rectangle.

To see how complex mappings transform the polar coordinate grid, you can compose the mapping in question with the exponential function. As an example, suppose we want to know what the mapping  $\sqrt{1+z^2}$  does to the unit disk. Then we could consider:

```
> conformal(sqrt(1+(exp(z)^2)),z=-3-Pi*I..1+Pi*I);
```



You can see that the result is contained in the right half plane, and some indication is

given of how it gets there, but more thought would be needed for a complete explanation.